

Lexical rules

1. Comments in Python begin with “#” and continue to the end of the line.
2. Lines containing only blanks, tabs, and comments are ignored entirely (they have no effect on indentation level). We call these blank lines hereafter.
3. Blanks and tabs may separate lexical tokens. They are required only to separate two tokens only when they could be interpreted differently if concatenated together (e.g. ”if” followed by ”x” could be interpreted as ”ifx” without intervening space). Blanks and tabs may not appear within individual lexical tokens (described in this section) other than string literals.
4. Blanks and tabs immediately at the beginnings of lines are significant. The amount of indentation of a line is the equivalent number of blanks at the beginning of that line. A tab following the equivalent of N blanks is equivalent to $8 - (N \bmod 8)$ blanks, so that each tab brings the equivalent number of blanks up to the next multiple of 8. Please see “Where to start” for more information about the rules for indentations.
5. If one line is indented more than the preceding non-blank line, it is taken to be preceded by an INDENT token. It is an error for the first non-blank line in a file to be indented.
6. If one line is indented less than the preceding non-blank line, it is taken to be preceded by enough DEDENT tokens to match all unmatched INDENT tokens introduced by preceding more-indented lines. The end of a file is preceded by enough DEDENT tokens to match all unmatched INDENT tokens.

Lexical tokens

NEWLINE: the new-line character

ENDMARKER: the end of file token

NUMBER: is defined by regular expression `[0-9]+`

ID: is defined by regular expression `[_a-zA-Z][_a-zA-Z0-9]*`

STRING: a squence of characters enclosed by either single or double quotes

Grammar notations

1. `{ rule }*` means any number of applications of *rule*, including no application (zero application) of *rule*.

2. { *rule* }+ means one or more applications of *rule*.
3. [*rule*] means *rule* is optional.
4. {(*or-separated-entities*) *rule* } means if one of the *or-separated-entities* is present, then *rule* applies.
5. Strings with single quotes around them are terminals.

Grammar rules

```
file_input: {NEWLINE | stmt}* ENDMARKER
```

```
stmt: simple_stmt | compound_stmt
```

```
simple_stmt: (print_stmt | assign_stmt) NEWLINE
```

```
print_stmt: 'print' [ testlist ]
```

```
assign_stmt: ID '=' test
```

```
compound_stmt: if_stmt | for_stmt
```

```
if_stmt: 'if' test ':' suite {'elif' test ':' suite}* ['else' ':' suite]
```

```
for_stmt: 'for' ID 'in' 'range' '(' testlist ')' ':' suite
```

```
suite: NEWLINE INDENT stmt+ DEDENT
```

```
testlist: test {',' test}*
```

```
test: or_test
```

```
or_test: and_test {'or' and_test}*
```

```
and_test: not_test {'and' not_test}*
```

```
not_test: 'not' not_test | comparison
```

```
comparison: arith_expr {comp_op arith_expr}*
```

```
comp_op: '<' | '>' | '==' | '>=' | '<=' | '<>' | '!='
```

```
arith_expr: term {('+' | '-') term}*
```

```
term: factor {('*' | '/' | '%' | '//') factor}*
```

```
factor: {'-'} factor | atom
atom: ID | NUMBER | STRING+ | '(' test ')'
```